Y. EL BORGI

Programme de tracé du polygone convexe associé a une loi symétrique

Les cahiers de l'analyse des données, tome 3, nº 2 (1978), p. 219-234

<http://www.numdam.org/item?id=CAD_1978__3_2_219_0>

Les Cahiers de l'Analyse des Données
Vol. III - 1978 - n° 2 - p. 219-234

PROGRAMME DE TRACÉ DU POLYGONE CONVEXE ASSOCIÉ A UNE LOI SYMÉTRIQUE [COSYM]

par Y. El Borgi (1)

1 *Objet du programme* : On rencontre souvent des tableaux rigoureusement symétriques, tableau de Burt, tableau de distances, tableau de proximités ; on a des tableaux carrés I x I dont la dissymétrie systématique est faible (matrices de confusion cf TI A n° 2 § 4 ; et J.P.Benzécri in Rev. Stat. Appl. : Vol XVII, pp 5-62, 1970) , en sorte qu'on peut avec avantage analyser le tableau symétrisé : $k^{sym}_{ii'} = (k_{ii'} + k_{i'i})/2$. Parfois (c'est le cas notamment pour les tableaux de flux) la diagonale (éléments k_{ii}) n'est pas connue; ou sa définition est problématique ; ce qui incite à considérer avec un tableau symétrique f_{II} d'autres tableaux f^{uv}_{II} construits à partir de celui-ci comme il est rappelé au § 2. L'ensemble de ces tableaux (qui fournissent tous les mêmes facteurs mais avec des v. p. différentes) constitue un polygone convexe plan, dont la représentation géométrique peut avoir l'intérêt pratique d'attirer l'attention sur un tableau f^{uv}_{II}, pour lequel le taux d'inertie afférent aux premiers facteurs est particulièrement élevé, d'où finalement une reconstitution améliorée des données à partir des facteurs. Le programme publié ici réalise le tracé du polygone convexe associé à une loi symétrique. Un autre article de ce même cahier (cf [TRAFIC] pp 203-218) donne un exemple de modèle de trafic conçu après l'analyse d'un tableau symétrique et examen du polygone associé.

2 *Rappel sur les correspondances symétriques* : (cf L'Analyse des D. TII B n° 9, [Corr. Sym.]). Les facteurs issus d'un tableau de correspondance symétrique f_{II} se répartissent en deux classes:

facteurs directs : $\forall\ i \in I : F_\alpha(i) = G_\alpha(i)$;

facteurs inverses : $\forall\ i \in I : F_\alpha(i) = -G_\alpha(i)$.

Cette notion dont le rôle dans l'interprétation est manifeste, dépend toutefois grandement du poids accordé à la diagonale du tableau. De façon précise, soit f_{II} une correspondance symétrique ; f_I sa loi marginale : à f_I sont associées sur I x I deux lois symétriques :

la loi diagonale, notée f^d_{II} ou f^{diag}_{II} :

$$\forall\ i, i' \in I : f^d_{ii'} = \delta_{ii'} \cdot f_i \quad ;$$

la loi produit, notée f^p_{II} ou f^{prod}_{II} :

$$\forall\ i, i' \in I : f^p_{ii'} = f_i \cdot f_{i'} \quad .$$

(1) *Docteur 3° cycle.*

Soit u , v deux nombres réels ; notons w = 1 - u - v. Considérons la loi :

$$f_{II}^{uv} = u\, f_{II} + v\, f_{II}^{d} + w\, f_{II}^{p} .$$

Physiquement la loi f_{II}^{uv} peut bien représenter la même réalité que f : on conçoit qu'en modifiant quelque peu les conditions d'une expérience ou la définition des données recueillies on puisse passer de f_{II} à f_{II}^{uv} sans cesser d'étudier le même objet.

La formule de reconstitution des données en fonction des facteurs nous fournit immédiatement l'analyse factorielle de f_{II}^{uv} à partir de celle de f_{II}. On a :

$$f_{ii'}^{uv} = f_i f_{i'} (1+ \Sigma_\alpha\{(u(\lambda_\alpha)^{1/2}\varepsilon_\alpha + v)\ \varphi_\alpha^{i}\, \varphi_\alpha^{i'}\})$$

ainsi les facteurs (facteurs normalisés de variance 1 φ_α^{I}) sont les mêmes pour f_{II} et f_{II}^{uv} , la valeur propre et la parité λ et ε changeant seulement suivant la formule :

$$(\lambda_\alpha^{uv})^{1/2} \,.\, \varepsilon_\alpha^{uv} = u\lambda_\alpha^{1/2}\ \varepsilon_\alpha + v\ ;$$

il apparaît à l'examen de cette formule que la parité $(\varepsilon_\alpha^{uv})$ peut varier avec u , v.

Pour une étude d'ensemble des lois f_{II}^{uv} ainsi associées à f_{II} , il convient de considérer tous les couples (u,v) de nombres (réels, non-nécessairement positifs) auxquels il correspond un tableau f_{II}^{uv} ne contenant que des nombres positifs : i.e. tels que :

$$\forall\ i, i' \in I :\ u\, f_{ii'} + v\,\delta_{ii'}\, f_i + (1 - u - v)\ f_i f_{i'} \geq 0.$$

Ce système d'inégalités definit dans le plan des (u,v) un polygone convexe P^+ que l'on appelle *polygone convexe associé* à la loi symétrique f_{II} ; et dont chaque point peut être identifié à la loi correspondante. Dans le plan du polygone on peut tracer deux droites passant par f_{II}^{p} et délimitant deux angles opposés par le sommet : l'un où tous les facteurs sont directs ; l'autre où tous les facteurs sont inverses. De plus est définie dans ce plan une métrique euclidienne naturelle :

$$d^2((u,v),\ (u',v')) = \Sigma_\alpha\{((\lambda_\alpha^{uv})^{1/2}\ \varepsilon^{uv} - (\lambda_\alpha^{u'v'})^{1/2}\ \varepsilon^{u'v'})^2\}$$

$$= \Sigma_\alpha\{((u - u')\,\lambda_\alpha^{1/2}\ \varepsilon_\alpha + (v - v'))^2\}\ ;$$

(où la somme est étendue au Card I - 1 facteurs). Muni de cette métrique le polygone convexe associé à f_{II} est le même que celui associé à l'un quelconque des f_{II}^{uv} : ce polygone caractérise la correspondance symétrique étudiée, indépendamment des choix faits quant à l'importance de la diagonale (v) ou des termes produits (1 - u - v). Se distinguent en particulier sur la figure les deux points f^1 et f^2 du polygone associé qui s'écartent le plus de la droite $f^{prod}\ f^{diag}$.

3 *Construction du polygone convexe associé à une loi symétrique* :

On a : $P^+ = \{f_{II}^{uv} \mid (u,v) \in R^2 ; \forall i, i' \in I : f_{ii'}^{uv} \geq 0\}$;

$P^+_{ex} = \{f_{II}^{uv} \mid (u,v) \in R^2 ; \forall i, i' \in I : i \neq i' \Rightarrow f_{ii'}^{uv} \geq 0\}$;

$P^+_{di} = \{f_{II}^{uv} \mid (u,v) \in R^2 ; \forall i \in I : f_{ii}^{uv} \geq 0\}$.

On a : $P^+ = P^+_{ex} \cap P^+_{di}$.

P^+_{di} est un polygone convexe ayant au plus $n = \mathrm{Card}\, I$ côtés (puisque c'est l'intersection de n demi-plans) et contenant à son intérieur les points $f_{II}^{diad} = f_{II}^{01}$ et $f_{II}^{prod} = f_{II}^{00}$.

P^+_{ex} est un angle de sommet f_{II}^{diag} contenant f_{II}^{prod} à son intérieur, puisqu'à chaque inégalité $f_{ii'}^{uv} \geq 0$, correspond un demi-plan contenant f_{II}^{prod} à son intérieur et dont la droite frontière passe par f_{II}^{diag} $(i \neq i' \Rightarrow f_{ii'}^{diag} = 0)$.

$P^+ = P^+_{ex} \cap P^+_{di}$ est donc un convexe limité à au plus $n+2$ droites ayant f_{II}^{diag} sur sa frontière et contenant f_{II}^{prod} à son intérieur.

Avec la distance rappelée ci-dessus, le produit scalaire de deux vecteurs $(u,v),(u',v')$, issus de l'origine $(0,0) = f_{II}^{00} = f_{II}^{prod}$ est donné par :

$$N((u,v), (u',v')) = \Sigma\{(u\,\lambda_\alpha^{1/2}\,\varepsilon_\alpha + v)(u'\lambda_\alpha^{1/2}\,\varepsilon_\alpha + v') \mid \alpha \in A\} ; \ *$$

D'où pour les vecteurs de base des axes Ou, Ov :

$$N((1,0), (1,0)) = \|(1,0)\|^2 = \Sigma\{\lambda_\alpha \mid \alpha \in A\} = \mathrm{trace}(f_{II}) = d^2(f_{II}^{prod}, f_{II});$$

$$N((0,1), (0,1)) = \|(0,1)\|^2 = n - 1 = d^2(f_{II}^{prod}, f_{II}^{diag}) ;$$

$$N((0,1), (1,0)) = \Sigma\{\lambda_\alpha^{1/2}\,\varepsilon_\alpha \mid \alpha \in A\}$$

et pour le cosinus de l'angle des axes Ou , Ov :

$$\cos\theta = \Sigma\{\lambda_\alpha^{1/2}\,\varepsilon_\alpha \mid \alpha \in A\}/((n-1)^{1/2}\,.\,\mathrm{trace}(f_{II})^{1/2}).$$

A chaque facteur φ_α est associée la droite d'équation :

$$(\lambda_\alpha^{uv})^{1/2}\,\varepsilon_\alpha^{uv} = u\,\lambda_\alpha^{1/2}\,\varepsilon_\alpha + v = 0 ;$$

cette droite, passant par $f_{II}^{prod} = f_{II}^{00}$, délimite dans P^+ deux régions l'une où φ_α est direct $((\lambda_\alpha^{uv})^{1/2}\,\varepsilon_\alpha^{uv} > 0)$ l'autre où φ_α est inverse ; l'ensemble de ces droites définit des secteurs angulaires ayant chacun sa signature propre : notamment, (comme on l'a annoncé au § 1) le secteur contenant f_{II}^{diag} où tous les facteurs sont directs, et son opposé, où tous sont inverses.

---

* *où A désigne l'ensemble des Card I − 1 facteurs de f_{II} autres que le facteur trivial constant.*

4 Le programme COSYM

Le programme se décompose en 11 parties :

A) Lecture des paramètres et du tableau des données, avec symétrisation de ce tableau si le paramètre NSYM est différent de zéro (cf § 5).

B) Calcul du tableau de fréquence à partir du tableau de contingence.

C) Calcul des coefficients des droites $f^{uv}_{ii} = 0$.

On a :

$$f^{uv}_{ii} = 0 \Leftrightarrow u(f_{ii} - f_i^2) + v(f_i - f_i^2) + f_i^2 = 0$$

$$\Leftrightarrow v = ((f_i^2 - f_{ii})/(f_i - f_i^2))u - f_i^2/(f_i - f_i^2)$$

On remarque que si pour $i \in I$, $f_{ii} = 0$, la droite $f^{uv}_{ii} = 0$ passe par le point (1,0).

D) Calcul des coefficients des deux droites formant l'angle P^+_{ex} :

Calculons l'équation de la droite $f^{uv}_{ii'} = 0$ pour $i \neq i'$;

$$f^{uv}_{ii'} = 0 \Leftrightarrow u(f_{ii'} - f_i f_{i'}) - v f_i f_{i'} + f_i f_{i'} = 0 \ ;$$

cette droite passe par le point (0,1). Il suffit donc de prendre le minimum des $C_{ii'} = 1/(1 - f_{ii'}/(f_i f_{i'}))$ positifs et le maximum des $C_{ii'}$ négatifs sur l'axe des u.



Dans le cas où les $C_{ii'}$ sont tous positifs ou tous négatifs on prendra leur minimum et leur maximum, et on a les deux cas suivants :





S'il existe i et i', $i \neq i'$, tels que $f_{ii'} = 0$ alors $C_{ii'} = 1$ et la droite : $u+v = 1$ est un des côtés de l'angle. Appelons C et D les deux valeurs de $C_{ii'}$ trouvées. Les deux côtés de l'angle P^+_{ex} auront comme équations respectives :

$$v = -(1/C)\ u + 1$$
$$v = -(1/D)\ u + 1$$

E) Détermination des sommets C du polygone convexe et du point f^1 le plus éloigné de la droite $f^{prod}\ f^{diag}$.

Il suffit de considérer l'ensemble des intersections de coordonnées (u,v) des droites déterminées en C) et D), et de ne garder que celles pour lesquelles :

$$\forall\ (i,i') \in I \times I\ :\ f^{uv}_{ii'} \geq 0.$$

Le carré de la distance d'un point de coordonnées (u,v) à la droite produit $f^{prod}\ f^{diag}$, i.e. à la droite passant par l'origine ($u = v = 0$) et le point f^{diag} ($u = 0$, $v = 1$) étant proportionnel à u^2 (et ceci quelque soit la métrique adoptée dans le plan des (u,v)) , le point f^1 du polygone convexe le plus éloigné de la droite précédente est le sommet de ce convexe, de coordonnées (u,v) dont la valeur absolue de l'abscisse $|u|$ est maximale.

F) Calcul et impression du tableau optimal f^1.

G) Calcul de l'angle des axes et de leur échelle.

H) Calcul et impression des caractéristiques du tableau optimal f^1 (valeurs propres - parités - pourcentages d'inertie)

I) Détermination des intersections F des droites des facteurs avec la droite $u = 1$ et de l'angle où tous les facteurs sont directs.

Les droites caractérisant les facteurs ayant pour équation $v = -u\ \lambda_\alpha^{1/2}\ \varepsilon_\alpha$ leurs intersections F avec la droite $u = 1$ a pour coordonnées : $u = 1$, $v = -\lambda_\alpha^{1/2}\ \varepsilon_\alpha$. En joignant l'origine aux deux points extrêmes de ces intersections, points qu'on note I, on obtient les droites délimitant l'angle où tous les facteurs sont directs (secteur délimité par les deux droites précédentes et contenant le point f^{diag} noté V) et l'angle opposé où ils sont tous inverses.

J) Passage à un repère orthonormé.

Soit $\vec{e}_1$ et $\vec{e}_2$ les vecteurs (1,0) et (0,1) et θ l'angle de ces deux vecteurs. La formule de passage des coordonnées (u,v) sur $(\vec{e}_1, \vec{e}_2)$ d'un vecteur, aux coordonnées (x,y) de ce vecteur sur un repère orthonormé dont le premier vecteur de base est $\vec{e}_1 / \|\vec{e}_1\|$ s'écrit :

$$x = u\ \|\vec{e}_1\| + v \|\vec{e}_2\|\ \cos\theta$$

$$y = v\ \|\vec{e}_2\|\ \sin\theta$$

$\|\vec{e}_1\|$, $\|\vec{e}_2\|$ et $\cos\theta$ étant donnés par les formules du §3. Dans le programme, $\|\vec{e}_1\|$ et $\|\vec{e}_2\|$ sont respectivement notés ECH1 et ECH2.

K) Représentation graphique.

On se sert du sous-programme NUAGE* pour tracer les sommets C du polygone convexe, les intersections F des droites des facteurs avec la droite $u = 1$, les intersections extrêmes étant notées I ; le point f^{diag}

* *sous-programme utilisé dans la version TABET du programme d'analyse factorielle des correspondances.*

(0,1) qui est aussi un sommet du polygone convexe est noté V, comme on l'a déjà dit ci-dessus, tandis que le point (1,0) qui est également représenté, et qui correspond au tableau initial est noté U.

Remarques

a) L'ensemble des points F, les deux points I et le point U sont alignés dans la représentation précédente.

b) Si le paramètre NVAL vaut zéro (cf § 5) le programme ne lit que le tableau de données, les valeurs propres et les parités n'étant pas connues ; dans ce cas les étapes G) à J) ne sont pas effectuées, et la représentation graphique (sans les points F et I) se fait en reportant les coordonnées (u,v) sur un graphique cartésien ordinaire.

5 Mode d'emploi et listage

5.1 Mode d'emploi : cartes en lecture :

1° carte : NEXPP = nombre d'expériences en I2
2° carte : carte paramètres, lue avec le format 40I2.

NA = nombre de lignes = nombre de colonnes
NSYM : paramètre de symétrisation :
si NSYM est différent de 0, on symétrise le tableau
NVAL : paramètre de lecture des valeurs propres et des parités associées ; on ne lit valeurs propres et parités que si NVAL est différent de 0.
NVA : paramètre associé au mode de lecture des valeurs propres ; si NVA vaut 1, on lit non pas les valeurs propres, mais la racine carrée de ces valeurs propres.

3° carte : Si NVAL est nul, cette carte est omise. Elle contient toutes les valeurs propres (si NVA $\neq 1$), ou leurs racines carrées (si NVA = 1), et est lue avec le format 10F8.5.
Si NA est supérieur à 11 (NA - 1 > 10), et plus petit ou égal à 21, il faudra deux cartes pour lire les valeurs propres.
Si NA est compris entre 22 et 31 (22 ≤ NA ≤ 31) il faudra trois cartes etc.

4° carte : Si NVAL est nul, cette carte est omise. Elle contient les signes des facteurs, et est lue avec le format 40F2.0.
Si NA est compris entre 42 et 81 (42 ≤ NA ≤ 81), il faudra deux cartes pour lire les signes des facteurs.
Si NA est compris entre 82 et 121, il faudra trois cartes etc....

5° carte : Elle contient le format des données, et se trouve juste avant le paquet de données. Elle est lue avec le format 20 A 4.

6° carte et cartes suivantes : le paquet de données de la première expérience lu avec le format précisé sur la 5° carte ; puis viennent les paquets (précédés de 4 cartes analogues aux cartes 2 à 5) correspondant aux expériences suivantes.

Remarque : Le programme est dimensionné pour analyser des tableaux dont la dimension NA n'excède pas 50. Pour des tableaux de taille plus importante, il suffira, si la valeur maximale NAMAX de NA au cours des analyses effectuées, dépasse 50 de modifier les trois cartes DIMENSION placées au début du programme.
Le tableau DEPAR sera dimensionné à NMAX x NMAX ;
les tableaux PB, EPS, VAL seront dimensionnés à NMAX
les tableaux A et B seront dimensionnés à NMAX + 2
les tableaux UA, VA, IIX, IIY et IDD seront dimensionnés à 3 * NMAX

5.2 Sorties du programme

En sortie, on imprime
- les valeurs des paramètres NA, NSYM, NVAL, NVA
- les valeurs propres (ou leur racine carrée si NVA = 1) ainsi que leur parité, cette impression n'ayant bien sûr lieu que si NVAL est non-nul.
- le format utilisé pour lire le tableau de données
- le tableau de données

- le tableau de données symétrisé (si NSYM est non-nul)
- le tableau symétrique des fréquences.
- les coefficients A et B des droites permettant de définir le polygone convexe associé à la correspondance étudiée.
- le tableau optimal, avec les valeurs associées de u et v.
- les caractéristiques de ce tableau optimal comparées avec celles du tableau initial (valeurs propres, parités, pourcentages d'inertie) si NVAL est non-nul.
- les coordonnées u et v des points intervenant dans le représentation graphique, ainsi que la nature de ces points indiqués par leur sigle (C,F,I,U,V, cf § 4)
- l'angle θ entre les vecteurs de base $\vec{e_1}$ (1,0) et $\vec{e_2}$ (0,1) des axes o u, o v, ainsi que les normes ECH1 et ECH2 de ces vecteurs, cette impression n'ayant bien sûr lieu que si NVAL est non-nul.
- les coordonnées des points intervenant dans la représentation graphique (ainsi que la nature de ces points) sur un système d'axes orthonormés dont le premier axe est l'axe o u, cette impression n'ayant lieu que si NVAL est non-nul.
- la représentation graphique.

Un exemple de sortie est fourni ci-dessous avec le listage du programme. Il est relatif à un tableau symétrique 8 x 8 issu d'une matrice de confusion symétrisée (avant d'être lue sur le programme).

5.3 *Listage du programme et exemple de sortie*

```
C
C     SUR UNE CORRESPONDANCE SYMETRIQUE OU SYMETRISEE
C     LE PROGRAMME COSYM DONNE LES PONDERATIONS U ET V  TELLES QU EN ANALYSANT
C     U*FII +V*FI*DI+(1-U-V)FI*FI ON A LE MAXIMUM D INERTIE POUR LE MINIMUM  DE
C     VALEURS PROPRES.
C      LE PROGRAMME FOURNIT  AUSSI  UN  GRAPHIQUE  OU  SONT  FIGURES  LES
C     SOMMETS  DU  POLYGONE  CONVEXE  ASSOCIE  A  LA  CORRESPONDANCE  ETUDIEE
C     AINSI  QUE  LES  INTERSECTIONS  DES  DROITES  DES  FACTEURS  AVEC  LA
C     DROITE  U=1 ,   CE  QUI  PERMET D AVOIR L ANGLE  DE  SOMMET L ORIGINE
C     ET  CONTENANT  FDIAG(U=0,V=1) OU TOUS LES FACTEURS  SONT DIRECTS
C
        DIMENSION DEPAR  (50,50)       ,PB(50),A( 52),B( 52)
      DIMENSION  UA(150) ,   VA(150)   ,IDD(150),IIX(150),IIY(150)
      DIMENSION FMT(20) ,  VAL(50),EPS(50)  ,TITRE(20)
      COMMON /TR/TITRE
      DATA IUU,IVV,IWW,IFF,IDI /1HU,1HV,1HC,1HF,1HI/
      DATA TITRE  /4HTRAC,4HE DU,4H POL,4HYGON,4HE CO,4HNVEX,4HE AS,4HSO
     1CI,4HE A ,4HUNE ,4HCORR,4HESPO,4HNDAN,4HCE S,4HYMET,4HRIQU,4HE    ,
     23*4H     /
 6000 FORMAT(////)
  800 FORMAT(//1X,20A4//)
      ETA0=1.E-07
      ETA5=-1.E-05
      READ100, NEXPP
      DO1285 NNAN=1,NEXPP
      PRINT1286
 1286 FORMAT(1H1)
      IDD(1)=IVV
      READ 100,     NA ,NSYM,NVAL,NVA
100   FORMAT (40I2)
      PRINT 900, NA ,NSYM,NVAL ,NVA
  900 FORMAT(10X,'NA=',I3, 5X,'NSYM=',I3,5X, 'NVAL=',I3 ,5X,'NVA=',I3//)
 160  NFAC=NA-1
      FACT=NFAC
      IF(NVAL.EQ.0)GOTO401
      READ 400,     (VAL(I),I=1,NFAC)
      READ 500,     (EPS(I),I=1,NFAC)
 400  FORMAT (10F8.5)
 500  FORMAT (40F2.0)
      PRINT 901, (I,EPS(I),VAL(I),I=1,NFAC)
  901 FORMAT(2X,'I',3X,'PARITE',3X,'VALEUR PROPRE'/(1X,I3,4X,F3.0,6X,F8.
     15))
  401 READ102,  FMT
 102  FORMAT (20A4)
      PRINT800,FMT
      PRINT403
  403 FORMAT(//1X,'TABLEAU INITIAL'/)
      NB = NA
      DO 103IA=1,NA
      READ FMT,    (DEPAR(IA,JB),JB=1,NB)
      PRINT150,    (DEPAR(IA,JB),JB=1,NB)
103   CONTINUE
C
C  SYMETRISATION  DU  TABLEAU
C
      IF(NSYM.EQ.0)GOTO402
      PRINT404
  404 FORMAT(//1X,'TABLEAU SYMETRISE'/)
      DO16IA=1,NFAC
```

```
      IABN=IA+1
      DO15JB=IABN,NA
      C=(DEPAR(IA,JB)+DEPAR(JB,IA))/2.
      DEPAR(IA,JB)=C
      DEPAR(JB,IA)=C
   15 CONTINUE
      PRINT150,   (DEPAR(IA,JB),JB=1,NB)
   16 CONTINUE
  402 TOTAL =0.
      DO 104 IA=1,NA
       DO 104 JB=1,NB
 104  TOTAL=TOTAL +DEPAR(IA,JB)
C
C     CALCUL DU TABLEAU DES FREQUENCES
C
      DO 106 JB=1,NB
      PB(JB) = 0
      DO 105 IA=1,NA
 105  PB (JB)=PB(JB)+DEPAR(IA,JB)
106   PB(JB)=PB(JB)/TOTAL
      PRINT149
  149 FORMAT(//1X,'TABLEAU DES FREQUENCES'/)
      DO110IA=1,NA
      DO111JB=1,NB
  111 DEPAR(IA,JB)=DEPAR(IA,JB)/TOTAL
  110 PRINT150,  ( DEPAR(IA,JB),JB=1,NB),PB(IA)
       PRINT150,  (PB(JB),JB=1,NB)
150   FORMAT( 1X,10E11.5)
      COEF1=1.E+09
      COEF2=-COEF1
      COEF3=0 .
      COEF4=0 .
C
C     CALCUL DESCOEF DES DROITES :FIJ=0
C
      NANA=NA-1
      DO303IA=1,NANA
       NBNB=IA+1
      DO302JB=NBNB,NA
      TTA=          1.-DEPAR(IA,JB)/(PB(IA)*PB(JB))
      TTT=SIGN(1.E+08,TTA)
      IF(ABS(TTA).GE.1.E-08)TTT=1./TTA
      IF (TTT.GT.0.)COEF1=AMIN1(COEF1,TTT)
      IF (TTT.LT.0.) COEF2=AMAX1(COEF2,TTT)
      IF(TTT.GT.0.) COEF3=AMAX1 (COEF3,TTT)
       IF(TTT.LT.0.) COEF4=AMIN1 (COEF4,TTT)
302   CONTINUE
 303  CONTINUE
      ETA1=COEF1-1.E+09
      ETA=ABS(ETA1)
      IF(ETA.LE.ETA0)COEF1=COEF4
      ETA1=COEF2+1.E+09
      ETA=ABS(ETA1)
      IF(ETA.LE.ETA0)COEF2=COEF3
      A(1)=-1./COEF1
      A(2)=-1./COEF2
      B(1)=1.
      B(2)=1.
C
C     CALCUL DES COEF DES DROITES:FII=0
```

```
C
      DO 202 IA=1,NA
      DEM=PB(IA)-PB(IA)*PB(IA)
      K=IA+2
      A(K)=PB(IA)*PB(IA)-DEPAR(IA,IA)
      B(K)=-PB(IA)*PB(IA)
      A(K)=A(K)/DEM
      B(K)=B(K)/DEM
 202  CONTINUE
      NBDROI=NA+2
      PRINT420
420   FORMAT(////1X,'COEFFICIENTS A B DES DROITES  Y=AX+B '/3X,'I',5X,
     2 'A',10X,'B'/)
      PRINT450,(I, A(I),B(I),I=1,NBDROI)
  450 FORMAT(1X,I3,2E12.5)
      UMAX=0.
      KMAX=1
      NCON=0
      NBD=NA+1
      DO12KP=1,NBD
      JC=KP+1
      AP=A(KP)
      BP=B(KP)
      DO2KS=JC,NBDROI
C
C  INTERSECTION  DES  DROITES  KP , KS
C
      AS=A(KS)
      BS=B(KS)
      IF(ABS(AS-AP).GE.1.E-08) GOTO3
      IF(ABS(BS-BP).GE.1.E-08)GOTO23
      PRINT151  ,KS,KP
       GOTO2
   23 PRINT152, KS,KP
      GOTO2
  151 FORMAT(1X,'LES DROITES',I3,' ET ',I3,' SONT CONFONDUES')
  152 FORMAT(1X,'LES DROITES',I3,' ET ',I3,' SONT PARALLELES')
    3 UAA=(BS-BP)/(AP-AS)
      VAA=(AP*BS-BP*AS)/(AP-AS)
C
C  L INTERSECTION  APPARTIENT  ELLE  AU  CONVEXE
C
      DO4K=1,NBDROI
      IF(K.EQ.KP.OR.K.EQ.KS) GOTO4
      TP=UAA*A(K)+B(K)-VAA
      IF(K.GE.3)TP=-TP
      IF(TP.LT. ETA5  )GOTO2
    4 CONTINUE
C
C L INTERSECTION  APPARTIENT  AU  CONVEXE
C
      NCON=NCON+1
      UA(NCON)=UAA
      VA(NCON)=VAA
      IF(NCON.GT.1)IDD(NCON)=IWW
      IF(ABS(UAA).LE.UMAX)GOTO2
      UMAX = ABS(UAA)
      KMAX=NCON
    2 CONTINUE
   12 CONTINUE
```

```
    8 FORMAT(1X,  3(I6,E15.8))
      U=UA(KMAX)
      V=VA(KMAX)
C
C   CALCUL  DU  TABLEAU  OPTIMAL (MAXIMUM D INERTIE  POUR  LE  MINIMUM  DE
C   VALEURS  PROPRES) , TABLEAU  ASSOCIE  AU*FII+V*FIIDIAG +(1.-U-V)*FIIPRODUCT
C,LE POINT(U,V) ETANT  LE POINT  DU  CONVEXE  LE  PLUS  ELOIGNE  DE  LA  DROITE
C  FIIDIAG(0,1) FIIPROD (0,0)
C
      PRINT6000
      PRINT410 ,U,V
  410 FORMAT(1X, 'TABLEAU OPTIMAL'/1X,'U=',E15.8,5X,'V=',E15.8/)
      W=1.-U-V
      DO 108 IA=1,NA
      DO 107 JB=1,NB
      PP = W*PB(IA)*PB(JB)
      PQ = 0.
      IF (IA.EQ.JB) PQ = V*PB(IA)
 107  DEPAR (IA,JB) = DEPAR(IA,JB)*U +PP+PQ
108   PRINT150,   (DEPAR(IA,JB),JB=1,NB),PB(IA)
      PRINT150,   (PB(JB),JB=1,NB)
      IF(NVAL.EQ.0)GOTO476
C
C     CALCUL DE L'ANGLE ET DES ECHELLES DES AXES
C
      SOM1=0.
      SOM2=0.
      DO 501 I=1,NFAC
      IF(NVA.EQ.1)GOTO17
      EPS(I)=EPS(I)*SQRT(VAL(I))
      GOTO18
   17 EPS(I)=EPS(I)*VAL(I)
      VAL(I)=VAL(I)*VAL(I)
   18 CONTINUE
      SOM1=SOM1+EPS(I)
501   SOM2=SOM2+VAL(I)
      ECH1=SQRT(SOM2)
      ECH2=SQRT(FACT)
      COCO=SOM1/(ECH1*ECH2)
      TETA= ACOS(COCO)
      PRINT6000
C
C  IMPRESSION  DES  CARACTERISTIQUES  DU  TABLEAU OPTIMAL
C
      PRINT6012
 6012 FORMAT(20X,'TABLEAU INITIAL',20X,'TABLEAU OPTIMAL'/)
      PRINT6013
 6013 FORMAT(2X,'I',2X,'RACINE CARREE DE',2X,'VALEUR PROPRE',2X,'POURCEN
     2TAGE',2X,'RACINE CARREE DE',2X,'VALEUR PROPRE',2X,'POURCENTAGE'/
     3,5X,'LA VALEUR PROPRE',17X,'D INERTIE',4X,'LA VALEUR POPPRE',17X,
     4'D INERTIE'/11X,'SIGNEE',40X,'SIGNEE'/)
 6011 FORMAT(1X,I3,E17.8,E15.8,E11.4,E17.8,E15.8,E13.4)
      DI= SOM2*U  *U  +2.*SOM1*V*U  +FACT*V*V
      DO6010I=1,NFAC
      D0=EPS(I)
      D=V  +EPS(I)*U
      E0=VAL(I)
      E=D**2
      F0=100.*E0/SOM2
      F=100.*E/DI
```

```
 6010 PRINT6011,I,D0,E0,F0,D,E,F
      PRINT6000
C
C  INTERSECTIONS  DES  DROITES  DES  FACTEURS  AVEC  LA  DROITE  U=1
C
      C1=1.
      C2=-1.
      IK1=NCON+1
      IK2=NCON+1
      DO606I=1,NFAC
      NCON=NCON+1
      IDD(NCON)=IFF
      UA(NCON)=1.
      VA(NCON)=-EPS(I)
      IF(EPS(I).LE.C2) GOTO607
      C2=EPS(I)
      IK2=NCON
      GOTO606
  607 IF(EPS(I).GE.C1) GOTO606
      IK1=NCON
      C1=EPS(I)
  606 CONTINUE
      IDD(IK1)=IDI
      IDD(IK2)=IDI
C
C  REPRESENTATION  GRAPHIQUE
C
  476 NCON=NCON+1
      IDD(NCON)=IUU
      UA(NCON)=1.
      VA(NCON)=0.
      PRINT407
  407 FORMAT(1X,'COORDONNEES U ET V DES SOMMETS C DU CONVEXE,  ET  DES
     1 INTERSECTIONS  F DES DROITES'/1X,'DES FACTEURS (DROITES PASSANT
     2 PAR L ORIGINE) AVEC   LA DROITE U=1.',/1X,'LES DROITES EXTREMES
     3 (PASSANT PAR O)DELIMANT  L ANGLE  CONTENANT FDIAG  (POINT V (U=0
     4,V=1))',/1X ,            'OU  TOUS LES FACTEURS SONT DIRECTS ONT
     5 LEURS INTERSECTIONS  AVEC U=1 NOTEES I'/
     6 1X,'SI NVAL = 0 ,LES POINTS I ET F NE SONT PAS MARQUES'/)
      DO14I=1,NCON
   14 PRINT11,I,UA(I),VA(I),IDD(I)
      IF(NVAL.EQ.0)GOTO477
      PRINT6000
      PRINT502,   COCO,TETA,ECH1,ECH2
502   FORMAT(1X,'COS TETA=',F10.5,2X,'TETA=',F10.5,2X,'ECH1=',F10.5,
     12X,'ECH2=',F10.5///)
      PRINT408
  408 FORMAT(1X,'COORDONNEES  SUR  DES  AXES  ORTHORNOMES'/)
      DO10I=1,NCON
      UA(I)=UA(I)*ECH1 +VA(I)*ECH2*COS(TETA)
      VA(I)=VA(I)*ECH2*SIN(TETA)
      PRINT11,I,UA(I),VA(I),IDD(I)
   11 FORMAT(1X,I6,2E15.8,1X,A3)
   10 CONTINUE
  477 CONTINUE
      MX=1
       CALL NUAGE (NCON,UA,VA,IDD,1,1,1,1,2,IIX,IIY,IDD)
 1285 CONTINUE
      STOP
      END
```

NA= 8 NSYM= 0 NVAL= 1 NVA= 0

I	PARITE	VALEUR PROPRE
1	1.	.43913
2	1.	.28056
3	1.	.24634
4	1.	.11037
5	1.	.09546
6	1.	.08154
7	1.	.04878

(3X,8F3.0)

TABLEAU INITIAL

.17500E 03	.70000E 02	.28000E 02	.15000E 02	.28000E 02	.49000E 02	.18000E 02	.21000E 02
.70000E 02	.23000E 03	.60000E 02	.29000E 02	.39000E 02	.27000E 02	.10000E 02	.14000E 02
.28000E 02	.60000E 02	.26300E 03	.78000E 02	.27000E 02	.20000E 02	.21000E 02	.25000E 02
.15000E 02	.29000E 02	.78000E 02	.24700E 03	.47000E 02	.32000E 02	.16000E 02	.10000E 02
.49000E 02	.27000E 02	.20000E 02	.32000E 02	.83000E 02	.18400E 03	.41000E 02	.14000E 02
.28000E 02	.39000E 02	.27000E 02	.47000E 02	.22500E 03	.83000E 02	.23000E 02	.10000E 02
.18000E 02	.10000E 02	.21000E 02	.16000E 02	.23000E 02	.41000E 02	.21500E 03	.91000E 02
.21000E 02	.14000E 02	.25000E 02	.10000E 02	.10000E 02	.14000E 02	.91000E 02	.20700E 03

TABLEAU DES FREQUENCES

.48103E-01	.19241E-01	.76965E-02	.41231E-02	.76965E-02	.13469E-01	.49478E-02	.57724E-02	.11105E 00
.19241E-01	.63222E-01	.16493E-01	.79714E-02	.10720E-01	.74217E-02	.27488E-02	.38483E-02	.13167E 00
.76965E-02	.16493E-01	.72292E-01	.21440E-01	.74217E-02	.54975E-02	.57724E-02	.68719E-02	.14349E 00
.41231E-02	.79714E-02	.21440E-01	.67894E-01	.12919E-01	.87960E-02	.43980E-02	.27488E-02	.13029E 00
.13469E-01	.74217E-02	.54975E-02	.87960E-02	.22815E-01	.50577E-01	.11270E-01	.38483E-02	.13249E 00
.76965E-02	.10720E-01	.74217E-02	.12919E-01	.61847E-01	.22815E-01	.63222E-02	.27488E-02	.12369E 00
.49478E-02	.27488E-02	.57724E-02	.43980E-02	.63222E-02	.11270E-01	.59098E-01	.25014E-01	.11957E 00
.57724E-02	.38483E-02	.68719E-02	.27488E-02	.27488E-02	.38483E-02	.25014E-01	.56899E-01	.10775E 00
.11105E 00	.13167E 00	.14349E 00	.13029E 00	.13249E 00	.12369E 00	.11957E 00	.10775E 00	

COEFFICIENTS A B DES DROITES Y=AX+B

I	A	B
1	-.82540E 00	.10000E 01
2	.20862E 01	.10000E 01
3	-.36236E 00	-.12492E 00
4	-.40134E 00	-.15163E 00
5	-.42071E 00	-.16752E 00
6	-.44935E 00	-.14981E 00
7	-.45773E-01	-.15272E 00
8	-.69325E-01	-.14115E 00
9	-.42557E 00	-.13581E 00
10	-.47107E 00	-.12076E 00

TABLEAU OPTIMAL
U= .14785538E 01 V= -.22040325E 00

.43464E-01	.24675E-01	.72664E-02	.23611E-02	.75816E-02	.16369E-01	.38877E-02	.54458E-02	.11105E 00
.24675E-01	.59982E-01	.19508E-01	.73576E-02	.11347E-01	.67690E-02	.22352E-07	.20274E-02	.13167E 00
.72664E-02	.19508E-01	.69949E-01	.26875E-01	.60658E-02	.35466E-02	.41058E-02	.61693E-02	.14349E 00
.23611E-02	.73576E-02	.26875E-01	.67286E-01	.14645E-01	.88450E-02	.24810E-02	.43999E-03	.13029E 00
.16116E-01	.64700E-02	.32208E-02	.85491E-02	.14901E-07	.70550E-01	.12574E-01	.20045E-02	.13249E 00
.78337E-02	.11646E-01	.63916E-02	.14941E-01	.87214E-01	.25204E-02	.55295E-02	.62350E-03	.12369E 00
.38877E-02	.22352E-07	.41058E-02	.24810E-02	.52580E-02	.12845E-01	.57335E-01	.33658E-01	.11957E 00
.54458E-02	.20274E-02	.61693E-02	.43999E-03	.37883E-03	.22492E-02	.33658E-01	.57383E-01	.10775E 00
.11105E 00	.13167E 00	.14349E 00	.13029E 00	.13249E 00	.12369E 00	.11957E 00	.10775E 00	

	TABLEAU INITIAL			TABLEAU OPTIMAL		
I	RACINE CARREE DE LA VALEUR PROPRE SIGNEE	VALEUR PROPRE	POURCENTAGE D INERTIE	RACINE CARREE DE LA VALEUR POPPRE SIGNEE	VALEUR PROPRE	POURCENTAGE D INERTIE
1	.66266662E 00	.43912703E 00	.3372E 02	.75938493E 00	.57666546E 00	.4309E 02
2	.52967441E 00	.28055501E 00	.2155E 02	.56274879E 00	.31668615E 00	.2367E 02
3	.49632955E 00	.24634302E 00	.1892E 02	.51344663E 00	.26362741E 00	.1970E 02
4	.33221680E 00	.11036801E 00	.8476E 01	.27079713E 00	.73331058E-01	.5480E 01
5	.30896926E 00	.95462024E-01	.7331E 01	.23642439E 00	.55896487E-01	.4177E 01
6	.28555739E 00	.81543028E-01	.6262E 01	.20180869E 00	.40726747E-01	.3043E 01
7	.22085738E 00	.48778001E-01	.3746E 01	.10614622E 00	.11267018E-01	.8420E 00

```
COORDONNEES U ET V DES SOMMETS C DU CONVEXE,  ET  DES   INTERSECTIONS  F DES DROITES
DES FACTEURS (DROITES PASSANT   PAR L ORIGINE) AVEC   LA DROITE U=1.
LES DROITES EXTREMES   (PASSANT PAR O)DELIMANT  L ANGLE  CONTENANT FDIAG  (POINT V (U=0 ,V=1))
OU  TOUS LES FACTEURS SONT DIRECTS ONT    LEURS INTERSECTIONS  AVEC U=1 NOTEES I
SI NVAL = 0 ,LES POINTS I ET F NE SONT PAS MARQUES

     1  .00000000E 00  .10000000E 01 V
     2  .14785538E 01 -.22040325E 00 C
     3 -.43826962E 00  .85690856E-01 C
     4  .55391710E-01 -.14499432E 00 C
     5  .38255379E-01 -.13878483E 00 C
     6  .49127859E 00 -.17521232E 00 C
     7  .10000000E 01 -.66266662E 00 I
     8  .10000000E 01 -.52967441E 00 F
     9  .10000000E 01 -.49632955E 00 F
    10  .10000000E 01 -.33221680E 00 F
    11  .10000000E 01 -.30896926E 00 F
    12  .10000000E 01 -.28555739E 00 F
    13  .10000000E 01 -.22085738E 00 I
    14  .10000000E 01  .00000000E 00 U


COS TETA=    .93943  TETA=    .34984  ECH1=   1.14113  ECH2=   2.64575


COORDONNEES  SUR  DES  AXES  ORTHORNOMES

     1  .24854946E 01  .90681446E 00 V
     2  .11394081E 01 -.19986480E 00 C
     3 -.28713781E 00  .77705681E-01 C
     4 -.29717350E 00 -.13148290E 00 C
     5 -.30129457E 00 -.12585205E 00 C
     6  .12512273E 00 -.15888506E 00 C
     7 -.50592518E 00 -.60091561E 00 I
     8 -.17537403E 00 -.48031640E 00 F
     9 -.92494965E-01 -.45007879E 00 F
    10  .31540549E 00 -.30125898E 00 F
    11  .37318712E 00 -.28017777E 00 F
    12  .43137723E 00 -.25894755E 00 F
    13  .59218872E 00 -.20027661E 00 I
    14  .11411285E 01  .00000000E 00 U
```

AXE HORIZONTAL(1)--AXE VERTICAL(1)--TITRE:TRACE DU POLYGONE CONVEXE ASSOCIE A UNE CORRESPONDANCE SYMETRIQUE

LARGEUR= 2.99142 HAUTEUR= 1.50773 -NOMBRE DE POINTS= 14